
    Working with Lua

Mix.install([
  {:lua, "~> 0.1.1"}
])
Setup
import Lua, only: [sigil_LUA: 2]
Lua
Executing Lua
code = ~LUA"""

"""

{value, %Lua{}} = Lua.eval!(code)
{[], #Lua<>}
Getting and Setting values
lua =
  Lua.new()
  |> Lua.set!([:dave], "Lucia")

{["Grohl", "Lucia"], %Lua{} = lua} =
  Lua.eval!(lua, ~LUA"""
  local original = dave;

  dave = "Grohl"

  return dave, original
  """)

Lua.get!(lua, [:dave])
"Grohl"
Exposing Elixir functions
Lua provides a module-based abstraction for defining APIs that can be called from Lua functions. Using use Lua.API with deflua defines APIs that are then loaded into the environment with the Lua.load_api/2 function.
In the example below, we load the my_print variadic function (a function that can take a variable number of arguments), and then call it from Lua.
defmodule Global do
  use Lua.API

  @variadic true
  deflua my_print(args) do
    IO.puts(Enum.join(args, " "))
  end
end

Lua.new()
|> Lua.load_api(Global)
|> Lua.eval!(~LUA"""
my_print("one", "two", "three")
""")
one two three
{[], #Lua<>}
There is quite a bit of flexibility for how functions can be defined. The module itself can define a scope under which all functions will be defined. In the previous example, no scope was defined, so functions were exposed in the globalscope.
Here we'll define a scope, "my.foo", and expose functions "bar" and "baz" underneath it
defmodule MyFooAPI do
  use Lua.API, scope: "my.foo"

  deflua bar(string) do
    "bar " <> string
  end

  deflua baz(string) do
    "baz " <> string
  end
end

{vals, %Lua{}} =
  Lua.new()
  |> Lua.load_api(MyFooAPI)
  |> Lua.eval!(~LUA"""
   return my.foo.bar("hi"), my.foo.baz("hello")
  """)

vals
["bar hi", "baz hello"]

Lua is an ergonomic interface to Luerl, aiming to be the best way to use Luerl from Elixir.
Features
	~LUA sigil for validating Lua code at compile-time
	deflua macro for exposing Elixir functions to Lua
	Improved error messages and sandboxing
	Deep setting/getting variables and state
	Excellent documentation and guides for working with Luerl

Lua the Elixir library vs Lua the language
When referring to this library, Lua will be stylized as a link.
References to Lua the language will be in plaintext and not linked.
Executing Lua
Lua can be run using the eval!/2 function
iex> {[4], _} = Lua.eval!("return 2 + 2")
Compile-time validation
Use the ~LUA sigil to parse and validate your Lua code at compile time
iex> import Lua, only: [sigil_LUA: 2]

#iex> {[4], _} = Lua.eval!(~LUA[return 2 +])
** (Lua.CompilerException) Failed to compile Lua!
Using the c modifier transforms your Lua code into a Lua.Chunk.t/0 at compile-time,
which will speed up execution at runtime since the Lua no longer needs to be parsed
iex> import Lua, only: [sigil_LUA: 2]
iex> {[4], _} = Lua.eval!(~LUA[return 2 + 2]c)
Exposing Elixir functions to Lua
The simplest way to expose an Elixir function to Lua is using the Lua.set!/3 function
import Lua, only: [sigil_LUA: 2]

lua = 
  Lua.set!(Lua.new(), [:sum], fn args ->
    [Enum.sum(args)]
  end)

{[10], _} = Lua.eval!(lua, ~LUA[return sum(1, 2, 3, 4)]c)
For easily expressing APIs, Lua provides the deflua macro for exposing Elixir functions to Lua
defmodule MyAPI do
  use Lua.API
      
  deflua double(v), do: 2 * v
end

import Lua, only: [sigil_LUA: 2]
    
lua = Lua.new() |> Lua.load_api(MyAPI)

{[10], _} = Lua.eval!(lua, ~LUA[return double(5)])
Calling Lua functions from Elixir
Lua can be used to expose complex functions written in Elixir. In some cases, you may want to call Lua functions from Elixir. This can
be achieved with the Lua.call_function!/3 function
defmodule MyAPI do
  use Lua.API, scope: "example"

  deflua foo(value), state do
    Lua.call_function!(state, [:string, :lower], [value])
  end
end

import Lua, only: [sigil_LUA: 2]

lua = Lua.new() |> Lua.load_api(MyAPI)

{["wow"], _} = Lua.eval!(lua, ~LUA[return example.foo("WOW")])
Modify Lua state from Elixir
You can also use Lua to modify the state of the lua environment inside your Elixir code. Imagine you have a queue module that you
want to implement in Elixir, with the queue stored in a global variable
defmodule Queue do
  use Lua.API, scope: "q"
  
  deflua push(v), state do
    # Pull out the global variable "my_queue" from lua
    queue = Lua.get!(state, [:my_queue])
    
    # Call the Lua function table.insert(table, value)
    {[], state} = Lua.call_function!(state, [:table, :insert], [queue, v])
    
    # Return the modified lua state with no return values
    {[], state}
  end
end

import Lua, only: [sigil_LUA: 2]

lua = Lua.new() |> Lua.load_api(Queue)

{[queue], _} =
  Lua.eval!(lua, """
  my_queue = {}

  q.push("first")
  q.push("second")

  return my_queue
  """)
  
["first", "second"] = Lua.Table.as_list(queue)
Accessing private state from Elixir
When building applications with Lua, you may find yourself in need of propagating extra context for use in your APIs. For instance, you may want to access information about the current user who executed the Lua script, an API key, or something else that is private and should not be available to the Lua code. For this, we have the Lua.put_private/3, Lua.get_private/2, and Lua.delete_private/2 functions.
For example, imagine you wanted to allow the user to access information about themselves
defmodule User do
  defstruct [:name]
end

defmodule UserAPI do
  use Lua.API, scope: "user"
  
  deflua name(), state do
    user = Lua.get_private!(state, :user) 
    
    {[user.name], state}
  end
end

user = %User{name: "Robert Virding"}

lua = Lua.new() |> Lua.put_private(:user, user) |> Lua.load_api(UserAPI)

{["Hello Robert Virding"], _lua} = Lua.eval!(lua, ~LUA"""
  return "Hello " .. user.name()
""")
This allows you to have simple, expressive APIs that access context that is unavailable to the Lua code.
Credits
Lua piggy-backs off of Robert Virding's Luerl project, which implements a Lua lexer, parser, and full-blown Lua virtual machine that runs inside the BEAM.

Calls a function in Lua's state
iex> {:ok, [ret], _lua} = Lua.call_function(Lua.new(), [:string, :lower], ["HELLO ROBERT"])
iex> ret
"hello robert"
References to functions can also be passed
iex> {[ref], lua} = Lua.eval!("return string.lower", decode: false)
iex> {:ok, [ret], _lua} = Lua.call_function(lua, ref, ["FUNCTION REF"])
iex> ret
"function ref"

The raising variant of call_function/3
This is also useful for executing Lua function's inside of Elixir APIs
defmodule MyAPI do
  use Lua.API, scope: "example"

  deflua foo(value), state do
    Lua.call_function!(state, [:string, :lower], [value])
  end
end

lua = Lua.new() |> Lua.load_api(MyAPI)

{["wow"], _} = Lua.eval!(lua, "return example.foo("WOW")")

Decodes a Lua value from its internal form
iex> {encoded, lua} = Lua.encode!(Lua.new(), %{a: 1})
iex> Lua.decode!(lua, encoded)
[{"a", 1}]

Deletes a key from private storage
iex> lua = Lua.new() |> Lua.put_private(:api_key, "1234")
iex> lua = Lua.delete_private(lua, :api_key)
iex> Lua.get_private(lua, :api_key)
:error

Encodes a Lua value into its internal form
iex> {encoded, _} = Lua.encode!(Lua.new(), %{a: 1})
iex> encoded
{:tref, 14}

Evaluates the Lua script, returning any returned values and the updated
Lua environment
iex> {[42], _} = Lua.eval!(Lua.new(), "return 42")
eval!/2 can also evaluate chunks by passing instead of a script. As a
performance optimization, it is recommended to call load_chunk!/2 if you
will be executing a chunk many times, but it is not necessary.
iex> {[4], _} = Lua.eval!(~LUA[return 2 + 2]c)
Options
	:decode - (default true) By default, all values returned from Lua scripts are decoded.        This may not be desirable if you need to modify a table reference or access a function call.
        Pass `decode: false` as an option to return encoded values

Gets a table value in Lua
iex> state = Lua.set!(Lua.new(), [:hello], "world")
iex> Lua.get!(state, [:hello])
"world"
When a value doesn't exist, it returns nil
iex> Lua.get!(Lua.new(), [:nope])
nil
It can also get nested values
iex> state = Lua.set!(Lua.new(), [:a, :b, :c], "nested")
iex> Lua.get!(state, [:a, :b, :c])
"nested"
Options
	:decode - (default true) - By default, values are decoded

Gets a private value in storage for use in Elixir functions
iex> lua = Lua.new() |> Lua.put_private(:api_key, "1234")
iex> Lua.get_private(lua, :api_key)
{:ok, "1234"}

Gets a private value in storage for use in Elixir functions, raises if the key doesn't exist
iex> lua = Lua.new() |> Lua.put_private(:api_key, "1234")
iex> Lua.get_private!(lua, :api_key)
"1234"

Inject functions written with the deflua macro into the Lua runtime.
See Lua.API for more information on writing api modules
Options
	:scope - (optional) scope, overriding whatever is provided in use Lua.API, scope: ...
	:data - (optional) - data to be passed to the Lua.API.install/3 callback

Loads string or Lua.Chunk.t/0 into state so that it can be
evaluated via eval!/2
Strings can be loaded as chunks, which are parsed and loaded
iex> {%Lua.Chunk{}, %Lua{}} = Lua.load_chunk!(Lua.new(), "return 2 + 2")
Or a pre-compiled chunk can be loaded as well. Loaded chunks will be marked as loaded,
otherwise they will be re-loaded everytime eval!/2 is called
iex> {%Lua.Chunk{}, %Lua{}} = Lua.load_chunk!(Lua.new(), ~LUA[return 2 + 2]c)

Loads a Lua file into the environment. Any values returned in the global
scope are thrown away.
Mimics the functionality of Lua's dofile

Initializes a Lua VM sandbox
iex> Lua.new()
By default, the following Lua functions are sandboxed.
	[:io]
	[:file]
	[:os, :execute]
	[:os, :exit]
	[:os, :getenv]
	[:os, :remove]
	[:os, :rename]
	[:os, :tmpname]
	[:package]
	[:load]
	[:loadfile]
	[:require]
	[:dofile]
	[:load]
	[:loadfile]
	[:loadstring]

To disable, use the sandboxed option, passing an empty list
iex> Lua.new(sandboxed: [])
Alternatively, you can pass your own list of functions to sandbox. This is equivalent to calling
Lua.sandbox/2.
iex> Lua.new(sandboxed: [[:os, :exit]])
Options
	:sandboxed - list of paths to be sandboxed, e.g. sandboxed: [[:require], [:os, :exit]]
	:exclude - list of paths to exclude from the sandbox, e.g. exclude: [[:require], [:package]]

Parses a chunk of Lua code into a Lua.Chunk.t/0, which then can
be loaded via load_chunk!/2 or run via eval!.
This function is particularly useful for checking Lua code for syntax
erorrs and warnings at runtime. If you would like to just load a chunk,
use load_chunk!/1 instead.
iex> {:ok, %Lua.Chunk{}} = Lua.parse_chunk("local foo = 1")
Errors found during parsing will be returned as a list of formatted strings
iex> Lua.parse_chunk("local foo =;")
{:error, ["Line 1: syntax error before: ';'"]}

Puts a private value in storage for use in Elixir functions
iex> lua = Lua.new() |> Lua.put_private(:api_key, "1234")

Sandboxes the given path, swapping out the implementation with
a function that raises when called
iex> lua = Lua.new(sandboxed: [])
iex> Lua.sandbox(lua, [:os, :exit])

Sets a table value in Lua. Nested keys will allocate
intermediate tables
iex> Lua.set!(Lua.new(), [:hello], "World")
It can also set nested values
iex> Lua.set!(Lua.new(), [:a, :b, :c], [])
These table values are availble in Lua scripts
iex> lua = Lua.set!(Lua.new(), [:a, :b, :c], "nested!")
iex> {result, _} = Lua.eval!(lua, "return a.b.c")
iex> result
["nested!"]
Lua.set!/3 can also be used to expose Elixir functions
iex> lua = Lua.set!(Lua.new(), [:sum], fn args -> [Enum.sum(args)] end)
iex> {[10], _lua} = Lua.eval!(lua, "return sum(1, 2, 3, 4)")
Functions can also take a second argument for the state of Lua
iex> lua =
...>   Lua.set!(Lua.new(), [:set_count], fn args, state ->
...>     {[], Lua.set!(state, :count, Enum.count(args))}
...>   end)
iex> {[3], _} = Lua.eval!(lua, "set_count(1, 2, 3); return count")

Sets the path patterns that Luerl will look in when requiring Lua scripts. For example,
if you store Lua files in your application's priv directory:
iex> lua = Lua.new(exclude: [[:package], [:require]])
iex> Lua.set_lua_paths(lua, ["myapp/priv/lua/?.lua", "myapp/lua/?/init.lua"])
Now you can use the Lua require function to import
these scripts
Warning
In order to use Lua.set_lua_paths/2, the following functions cannot be sandboxed:
	[:package]
	[:require]

By default these are sandboxed, see the :exclude option in Lua.new/1 to allow them.

Write Lua code that is parsed at compile-time.
iex> ~LUA"return 2 + 2"
"return 2 + 2"
If the code cannot be lexed and parsed, it raises a Lua.CompilerException
#iex> ~LUA":not_lua"
** (Lua.CompilerException) Failed to compile Lua!
As an optimization, the c modifier can be used to return a pre-compiled Lua chunk
iex> ~LUA"return 2 + 2"c

Defines the Behaviour for defining a Lua API
To create a module that exports functions to the global scope
defmodule MyAPI do
  use Lua.API

  # Can be called via `print("hi")` in lua
  deflua print(msg), do: IO.puts msg
end
Optionally, you can provide a scope
defmodule SpecificAPI do
  use Lua.API, scope: "namespace.domain"

  # Can be called via `namespace.domain.foo(5)` in lua
  deflua foo(v), do: v
end
You can access Lua state
defmodule State do
  use Lua.API

  deflua bar(name), state do
    # Pull's the value of `number` out of state
    val = Lua.get!(state, [:number])

    2 * val
  end
end
Regular functions are not exported
defmodule SpecificAPI do
  use Lua.API

  # Won't be exposed
  def baz(v), do: v
end
Installing an API
A Lua.API can provide an optional install/3 callback, which
can run arbitrary Lua code or change the Lua state in any way.
An install/3 callback takes a Lua.t/0 and should either return a
Lua script to be evaluated, a Lua.Chunk.t/0, or return a new Lua.t/0
defmodule WithInstall do
  use Lua.API, scope: "install"

  @impl Lua.API
  def install(lua, _scope, _data) do
    Lua.set!(lua, [:foo], "bar")
  end
end
If you don't need to write Elixir, but want to execute some Lua
to setup global variables, modify state, or expose some additonal
APIs, you can simply return a Lua chunk directly using the c modifier
on Lua.sigil_LUA/2
defmodule WithLua do
  use Lua.API, scope: "whoa"

  import Lua

  @impl Lua.API
  def install(_lua, _scope, _data) do
    ~LUA[print("Hello at install time!")]c
  end
end

          @callback install(Lua.t(), scope_def(), any()) :: Lua.t() | Lua.Chunk.t() | String.t()

Defines a function that can be exposed in Lua through Lua.load_api/3
deflua add_two(number) do
  number + 2
end
Accessing state
Sometimes, you may want to access or modify the Lua environment in a deflua. This
can be done by using the following syntax
deflua get_value(key), state do
  # Access the Lua environment
  Lua.get!(lua, [key])
end
To modify and return new state, return a tuple
deflua set_value(key, value), state do
  # Return nothing but modify the state
  {[], Lua.set!(lua, [key])}
end
Variadic functions
Technically, all Lua functions are variadic, which means they can receive
a variable number of arguments. As a convenience, Lua applies your arguments to
deflua functions so that they can be written in idiomatic Elixir.
If you need to handle variadic arguments, annotate the function with the @variadic
module attribute.
@variadic true
deflua print(args) do
  IO.puts(Enum.join(args, " "))
end
@variadic behavior
When using the @variadic attribute, note that it is per-function. Lua will
reset this attribute after every function definition, so there is no need to
manually reset it yourself

Raises a runtime exception inside an API function, displaying contextual
information about where the exception was raised.

A pre-compiled chunk of Lua code
that can be executed at a future point in time

In Lua, tables are the fundamental datastructure,
which are used both as associative arrays (maps), and arrays (lists).
Lua.Table provides some utilities for working with Lua tables when passed back to
Elixir.

        Converts a Lua table into a list. Assumes that the
table is correctly ordered.

        Converts a Lua table into a map



    


    
      
        as_string(table, opts \\ [])

      


        Converts a Lua table into a string representing
a Lua table literal

        Converts a Lua table into more "native" feeling lists and
maps, deeply traversing any sub-tables.

Converts a Lua table into a list. Assumes that the
table is correctly ordered.
iex> Lua.Table.as_list([{1, "a"}, {2, "b"}, {3, "c"}])
["a", "b", "c"]
To ensure the list is ordered, you can pass the :sort option
iex> Lua.Table.as_list([{2, "b"}, {1, "a"}, {3, "c"}])
["b", "a", "c"]

iex> Lua.Table.as_list([{2, "b"}, {1, "a"}, {3, "c"}], sort: true)
["a", "b", "c"]

Converts a Lua table into a map
iex> Lua.Table.as_map([{"a", 1}, {"b", 2}])
%{"a" => 1, "b" => 2}

  



    

  
    
      
    
    
      as_string(table, opts \\ [])



        
          
        

    

  


  

Converts a Lua table into a string representing
a Lua table literal
iex> Lua.Table.as_string([{"a", 1}, {"b", 2}])
"{a = 1, b = 2}"
Lists that "look" like Lua tables are treated as lists
iex> Lua.Table.as_string([{1, "foo"}, {2, "bar"}])
~S[{"foo", "bar"}]
Lists are treated as lists
iex> Lua.Table.as_string(["a", "b", "c"])
~S[{"a", "b", "c"}]
Regular maps are always treated as tables
iex> Lua.Table.as_string(%{1 => "foo", "bar" => "baz"})
~S<{[1] = "foo", bar = "baz"}>
Options
	:formatter - A 2-arity function used to format values before serialization. The key and value
are passed as arguments. If there is no key, it will default to nil.


  



  
    
      
    
    
      deep_cast(value)



        
          
        

    

  


  

Converts a Lua table into more "native" feeling lists and
maps, deeply traversing any sub-tables.
It uses the heuristic that maps with integer keys starting
as 1 will be auto-cast into lists
iex> Lua.Table.deep_cast([{"a", 1}, {"b", [{1, 3}, {2, 4}]}])
%{"a" => 1, "b" => [3, 4]}

  


        

      


  

    
Lua.CompilerException exception

        message(compiler_exception)

      


        Callback implementation for Exception.message/1.

Callback implementation for Exception.message/1.

  


        

      


  

    
Lua.RuntimeException exception

